Datatypes

1. Una forma simple de declarar un nuevo tipo de dato

Podemos definir un tipo de dato nuevo llamado `fruta` que consista de tres valores posibles: `Pera`, `Naranja`, `Manzana`, de la siguiente manera:

```
- datatype fruta = Pera | Naranja | Manzana;
datatype fruta = Manzana | Naranja | Pera
```

Una vez definido este nuevo tipo de dato, podemos usarlo en una función u otra expresión. Por ejemplo, podríamos definir:

```
- fun esPera(x) = (x = Pera);
val esPera = fn : fruta ->bool
```

Esta función devuelve verdadero si se le pasa como argumento el valor `Pera` y falso para cualquier otro valor de tipo `fruta`. Es un error pasarle como argumento cualquier cosa que no sea alguno de los constructores del tipo de dato `fruta`.

Veamos las siguientes respuestas:

```
- esPera(Manzana);
val it = false : bool
- esPera(Naranja);
val it = false : bool
- esPera(Banana);
stdIn:7.8-7.14 Error: unbound variable or constructor: Banana
```

Esta última respuesta nos indica que `Banana` no es un argumento válido para la función `esPera`.

1.1. Usando expresiones constructoras para definir nuevos tipos de datos

La forma general de una declaración de tipo es:

`datatype (`<*lista-de-parámetros*>`)` <*identificador-nuevo-tipo*> `=`
<*1era-expresión-constructora*> `|`
<*2da-expresión-constructora*> `|`
...
<*últ-expresión-constructora*>

Una expresión constructora consiste en el nombre del constructor, la palabra `of` y una expresión de tipo.

Un ejemplo simple:

```
- datatype b = Banana of int
```
datatype b = Banana of int

Un ejemplo más complejo:

```
- datatype ('a, 'b) elemento =
            P of 'a * 'b |
            S of 'a;
```

```
datatype ('a,'b) elemento = P of 'a * 'b | S of 'a
```

Este tipo de dato representa la unión entre elementos simples y pares ordenados.

Supongamos ahora que tomamos una lista de `(string, int) elemento` y queremos sumar los enteros de las segundas componentes de aquellos elementos que tienen esta segunda componente.

La función `sumaLaLista` hace lo que queremos:

```
- fun sumaLaLista(nil) = 0
| sumaLaLista(S(x)::L) = sumaLaLista(L)
| sumaLaLista(P(x,y)::L) = y + sumaLaLista(L);
val sumaLaLista = fn : ('a,int) elemento list -> int
```

1.2. Tipo de dato recursivo

En muchos casos, necesitamos aplicar constructores de datos de manera recursiva para construir expresiones arbitrariamente grandes.

Un árbol binario etiquetado se define recursivamente como sigue:

```
- datatype 'etiqueta arbolbin =
            Vacio |
            Nodo of 'etiqueta * 'etiqueta arbolbin * 'etiqueta arbolbin;
datatype 'a arbolbin = Nodo of 'a * 'a arbolbin * 'a arbolbin | Vacio
```

Un ejemplo de un valor de tipo `string arbolbin`:

```
- Nodo("Sabrina",
        Nodo("Pedro",
                Nodo("Juan", Vacio, Vacio),
                Nodo("Natalia", Vacio, Vacio)
        ),
        Nodo("Roberto", Vacio, Vacio)
);
```

Y representa el siguiente árbol binario: